INTERFACING A PAPER TAPE READER TO THE SWTPC 6800

By Michael P. Burton

Owners of the Southwest Technical Products Corporation 6800 Computer often use TV typewriters as their serial control device, instead of using an ASR33 teletype. This leaves them without a method of reading paper tape programs.

Faced with this situation, I had two alternatives. I could get my paper tape version of TINY BASIC 6800 listed and type it in by hand, or I could purchase the Oliver Audio Engineering OP-80A paper tape reader and a SWTPC MP-L parallel interface board and d o things properly. After spending five hours typing in TINY BASIC, I decided that once was enough, so I purchased the reader and interface.

Construction of the paper tape reader is very easy. The reader should be jumpered so that point A goes ACK. There are no modifications necessary to make the OP-80A compatible with. the SWTPC MP-L parallel interface.

Construction of the MP-L parallel interface is also an easy task. Jumper the board for maskable interrupt operation (A to IRQ and B to IRQ).Do not insert the small indexing plug into the Molex edge connector at the top of the board, and do not cut off the corresponding index pin of the male Molex input connector. Solder a jumper between the 5 V regulated side of IC 2 and the index point of the Molex edge connector.

Connector wiring of the OP-80A to the parallel interface is shown in the figure. Note that the index pin of the interface board is used to supply power to the OP-80A. Since the tape reader draws a maximum of 175 mA current, the regulated 5 V required cab be obtained from the interface board. Also note that S1 and S2 are being used as visual monitors of the data read line (RDA) and data acknowledge line (ACK).An ACK signal is used to reset the RDA line, as the signal source (CB2 of the PIA) is a normally HI signal when used as an output.



The actual programming of the MP-L for operation with the OP-80A may be on an interrupt or non-interrupt basis. The example shown is two hand-assembled, non-interrupt driven subroutines that set up the peripheral interface adapter and read one frame of the paper tape. Note that the PIA address used is 8000 8003, which is I/O slot zero on the SWTPC Mother Board. Information concerning interrupt programming is set forth in the MC6820 Peripheral Interface Adapter data sheets, and in the Hardware section of the SWTPC System Documentation Notebook..

```
                *PIASUP - PIA SETUP ROUTINE
0F00                    ORG     $0F00
0F00 CE 8000 PIASUP     LDX     #$8000          PIA ADDRESS
0F03 86 36              LDA A   #$36            KNOCK DOWN C2(RESET C1)
0F05 A7 03              STA A   3,X             SET CRB
0F07 4F                 CLR A
0F08 A7 01              STA A   1,X             RESET CRA
0F0A A7 02              STA A   2,X             SET DDRB = ALL INPUTS
0F0C 43                 COM A
0F0D A7 00              STA A   0,X             SET DDRA = ALL OUTPUTS
0F0F 86 3E              LDA A   #$3E            C1 SET ON A 0 TO 1, C2=1
0F11 A7 03              STA A   3,X             SET CRB
0F13 86 04              LDA A   #$04            DUMMY FOR CRA
0F15 A7 01              STA A   1,X             SET CRA
0F17 39                 RTS                     AND LEAVE...
                *
0F18 0001       DATA    RMB     1               DATA AREA
                *
                *PIADAT - GET DATA FROM PIA
0F19 CE 8000 PIADAT     LDX     #$8000          PIA ADDRESS
0F1C A6 00              LDA A   0,X             READ ADR AND
0F1E A6 02              LDA A   2,X             BDR TO CLEAR CR FLAGS
0F20 86 3E              LDA A   #$3E            C1 SET ON A 0 TO 1, C2=1
0F22 A7 03              STA A   3,X             SET CRB
0F24 A6 03      DATLOP  LDA A   3,X             GET C1 FLAG(BIT 7)
0F26 2B 02              BMI     GETDAT          DATA PRESENT?
0F28 20 FA              BRA     DATLOP          NOT YET.  KEEP TRYING.
0F2A A6 02      GETDAT  LDA A   2,X             YES.  GET THE DATA
0F2C B7 0F18            STA A   DATA            AND SAVE IT.
0F2F 86 36              LDA A   #$36            KNOCK DOWN C2(RESET C1)
0F31 A7 03              STA A   3,X             SET CRB
0F33 39                 RTS                     AND LEAVE...
```